
A note about SAS:


SAS works primarily through a command language interface.  We have sample command files that can be run to generate the output shown here, or modified to try other options.

SAS programs are written in the SAS programming language. A series of SAS statements together form a SAS program. SAS statements start with a key word, and end with a semicolon. SAS statements are strung together to form Data Steps and Proc Steps. A data step creates a new data set. Data steps are necessary to modify variables, carry out recodes of variables, etc. Proc steps are used to carry out analyses, using SAS procedures. Each procedure will have a number of options and statements that modify the action of the procedure. Some key SAS concepts are defined below:

SAS Command files:

SAS command files are SAS programs that do a series of steps. They can be used to set up data, run procs on existing data sets, or both. You should save your SAS commands with the .sas extension. For example:

key_concepts.sas

To Open SAS Command files:

1. Enhanced Program Editor Window: (We will be use this for the demonstration)
First open SAS. Be sure your current window is the Enhanced Program Editor Window. Go to File > Open Program. . . and browse until you find the name of the command file (e.g., key_concepts.sas), then double-click on it. You will see color-coded commands in SAS, with lines separating portions of the program. To run a specific portion of the commands, highlight that portion and click on the running-person icon in the menus. Be sure to highlight all of the commands including “run;”. If you do not highlight the semicolon after “run”, the commands will not be executed. 

To run all of the commands in the Enhanced Program Editor at once, simply click on the running-person icon without selecting any code. You should periodically save the commands in your Enhanced Editor Window, because SAS will not automatically save them for you. If you inadvertantly close the Enhanced Editor Window, you will need to go to View>Enhanced Editor…and then re-open your command file.

You may have several programs open in the Enhanced Program Editor at once. They will all be listed at the bottom of your SAS window. If you pass your mouse over the files listed at the bottom of your SAS window, the names of the files will be displayed. You should periodically save the commands in your Enhanced Editor Window, because SAS will not save them for you.

2. Program Editor Window: (Another way to work with SAS commands)
First open SAS. Open the Program Editor Window (Select View > Program Editor). Then go to File > Open Program. . ., browse to the name of the file (e.g., key_concepts.sas) and double-click on it. You will see color-coded commands in SAS. To run a specific portion of the commands, highlight that portion and click on the running-person icon in the menus. Be sure to highlight all of the commands including “run;”. If you do not highlight the semicolon after “run”, the commands will not execute. 

To run all of the commands in the Program Editor Window at once, select all of the code, and then click on the run icon. If you simply click on the running-person icon without selecting any code, all of the code will disappear from the Program Editor Window. To get it back, go to the Command Dialog Box at the top left of the SAS desktop and type:

recall

and hit return. Your previously submitted commands will be recalled to the Program Editor window. If you inadvertantly close the Program Editor Window, you can recover your program by going to View>Program Editor.


You may have only one program open at a time in the Program Editor Window. You should periodically save your commands, because SAS will not save them for you.

3. Double-Click on a SAS command file: 


This method should open your command file in the SAS Enhanced Program Editor Window. It will work if you have SAS installed on your personal computer, but may not work at the computing lab. If it does not, use the first or second method above.

Editing Commands in the Enhanced Program Editor Window or Program Editor Window:

You can edit commands in the either the Enhanced Program Editor or Program Editor. You can use the usual windows commands in these windows, including Copy (Ctrl-C), Cut (Ctrl-X), and Paste (Ctrl-V) by utilizing the icons at the top of the SAS menus, or by using the shortcut keys. There is also an Undo (Ctrl-Z) button at the top of the menus. You can Undo a number of times, to get back to an earlier version of your commands. You should periodically save your SAS commands, because SAS will not automatically save them for you.  

Break Key: The Break key is an exclamation point within a circle in the menus at the top of the SAS menus. It will interrupt commands that you wish to stop before they complete executing.

SAS Log Window: 

This window contains a cumulative log of all commands and any errors, warnings or notes that SAS produces upon parsing your syntax. Retain this window to help de-bug your program. If you have an error in the log, it will show up in Red.  An error will stop the particular data or proc step you are running, until it is fixed. Check the log for the first occurrence of an error, and fix that, and then rerun your program from the Program Editor Window. Often, fixing a single error (such as the exclusion of a semicolon) will allow a program to run without any errors. It is good practice to check your log each time you run a Data Step or Proc Step to be sure that the program executed properly. 

Warning messages will show up as green in the log. You may not edit the information in the SAS Log Window.

SAS Output Window: 

This window contains any output from your SAS runs. You can print the entire window, by going to the output window and selecting File…Print. You can also select portions of the output to print, by going to the Results sidepanel and selecting one portion of the output to print. You can also simply highlight portions of the output window to print. Be careful that you stop your selection on a line that has printing on it, or SAS will give you an error message. You cannot edit information in the SAS Output Window.

Crummy looking SAS output can be made to look better by changing the default font that is used with an options statement. This problem usually occurs if you are trying to print SAS output from a computer on which SAS is not installed, and therefore it does not have the SAS font available.
options formchar="|----|+|---+=|-/\<>*";
Comments in SAS Code:

There are two basic kinds of comments in SAS. Those that start with a /* and end with */ can go over more than one line, and can knock out whole blocks of code, or can be inserted within a given line of code. Those that start with an * and end with a semicolon simply knock out a single SAS statement. Comments show up in Green font in the SAS Program Editor and Enhanced Program  Editor. Comments are used throughout the command files for the labs to document the SAS code, and to give you an idea of what is being done. Comments will not be executed by SAS.

/* One type of comment */

/*****************************************************

 Same as above, but with more asterisks

 All code between the “gates” will be excluded

******************************************************/

*Another type of comment;

Libraries in SAS: 


Libraries are usually folders that contain SAS data sets and other SAS files, such as formats catalogs. The libraries WORK, SASUSER and SASHELP are automatically defined each time you open SAS. The WORK library is the default location for any temporary data sets you create during your current SAS run. Data sets stored in WORK will be automatically deleted when you are done with your current session.

A libname statement defines a folder where permanent SAS data sets, SAS formats catalogs and other SAS files are stored. A libname statement must be submitted before a permanent SAS data set within a folder can be read. You must enclose the path to the folder in quotes. Single or double quotes can be used. Statements in SAS can be in lower or upper case.  

libname sasdata1 V6 "c:\temp\sasdata1\";

libname sasdata2 V9 "c:\temp\sasdata2\";

In this case we define sasdata1 as a folder containing version 6 SAS data sets (specified by using the V6 option). We define sasdata2 as a folder containing version 9 SAS data sets. 

Permanent SAS data sets: 

The following version 9 SAS data sets are included in the sasdata2 library: 

bank.sas7bdat

baseball.sas7bdat

business.sas7bdat

iris.sas7bdat

tecumseh.sas7bdat

You can use these version 9 SAS data sets by simply referring to each data set by its two-level name (library.dsn). Do not include the file extension in the data set name. Note that the two-level data set name must be specified for each procedure. It does not become the default by simply using it once:

/*USE PERMANENT SAS DATA SETS*/
libname sasdata2 V9 "c:\temp\sasdata2\";
proc print data=sasdata2.business;

run;

proc means data=sasdata2.business;

run;

proc contents data=sasdata2.business;

run;

To set up a default SAS data set, submit an options statement, to tell SAS to set the “last” data set to be the one you specify, as shown below;

options _last_ = sasdata2.business;

proc means;

run;

proc contents; 
run;
The contents of all SAS data sets in the SASDATA2 library can be displayed with the following syntax:

proc contents data=sasdata2._all_;

run;

The sasdata2 library also contains a SAS transport file, bank.xpt. You can use the Proc Copy command to copy the file from the transport format into a version 9 SAS data set. 

/*USE A TRANSPORT FORMAT FILE

  NOTE: THE ENTIRE PATH AND 

        FILE NAME ARE SPECIFIED*/

libname trans xport "c:\temp\sasdata2\bank.xpt";

proc copy in=trans out=sasdata2;

run;

The sasdata1 library contains these Version 6 SAS data sets:


FITNESS.SD2


GPA.SD2


MARCH.SD2

You can utilize the version 6 SAS data sets in the sasdata1 forder, by  first submitting a libname statement to define the library where the data sets are stored, which tells SAS to use the Version 6 engine when reading these files, and then specifying each data set to use with a two-level name, as shown below:

libname sasdata1 V6 "c:\temp\sasdata1\";

proc print data=sasdata1.gpa;

run;

proc means data=sasdata1.gpa;

run;

proc contents data=sasdata1.gpa;

run;

Formats Catalogs: 

The library SASDATA1 also contains the version 6 formats catalog:

 
formats.sc2
And the SASDATA2 library contains the version 9 formats catalog:


formats.sas7bcat

Using SAS Formats: 

You can utilize the formats in these formats catalog or in any format catalog  within the SASDATA1, SASDATA2, or WORK library by specifiying an options fmtsearch= statement to tell SAS where to search for formats. SAS will search for formats in these libraries, in the order you specify. If you have the same format name in two formats libraries, it will choose the one in the library mentioned first in the fmtsearch option.The options nofmterr; statement tells SAS not to give you an error, if there is a problem with a format for a given variable or variables. (Note: These options are given at the start of the command file.)
options fmtsearch=(sasdata1 sasdata2 work);

options nofmterr;

proc freq data=sasdata1.owen;

  tables sex;

  format sex sexfmt.; run;
Migrate files from SAS version 6 to version 9:

To convert all version 6 SAS data sets in the sasdata1 folder to version 9 files, you can use Proc Migrate:

proc migrate in=sasdata1 out=sasdata2;

run;

The folder sasdata2 must already exist, and cannot have SAS files with the same names as the ones you are trying to migrate already in it. If there is, the data set will not be migrated.

The formats catalog will not be migrated from version 6 to version 9, because a formats catalog already exists in the sasdata2 library.
Data Step:

A data step is used to create a new data set, either by reading in raw data, or by modifying an existing SAS data set. New variables, or recodes of existing variables must be created using a data step. The data step begins with the keyword data in the data statement. The data step is completed with a run statement. Data sets can be temporary or permanent. A temporary data set is only available during a specific SAS session, and will be lost when that session is over, whereas a permanent data set will be available for future use. Temporary SAS data sets may be specified with only one name, as shown below, or they can have a two-level name (e.g. WORK.PULSE). Temporary data sets are automatically stored in the WORK library. Permanent SAS data sets must have two-level names, including both the library and the data set name.

The data step below tells SAS to create a temporary data set called BANK from the permanent data set, sasdata2.bank. The set statement in the data step below tells SAS the permanent data set to read from. New variables are created by recoding values of the original variables.

data bank;

  set sasdata2.bank;

  if jobcat in (1,3,5,7) then trainee = 0;

  if jobcat in (2,4,6) then trainee = 1;

run;
/*THE RUN STATEMENT CLOSES THE DATA SET*/
Missing Value Codes:

The missing value code for a numeric variable by default is a period (.) in SAS. All missing values will be excluded from analyses, such as a regression analysis, by default. The missing value code for a character variable is “  “ (quote blank quote).

Proc Steps:

A Proc step carries out a given procedure. Each procedure will have its own specific statements and options. To get help on a given procedure, go to the SAS command dialog box at the top left of your SAS session and type 

Help Procname <return>

For example, to get help on Proc Reg, you would type:

help reg
Temporary data sets can be referred to by their full name, e.g., WORK.BANK, or by their one-level name, e.g. BANK. Permanent SAS data sets must be referred to by their two-level name. By default, the most recently created data set from the current session is used, if no data set is specified for a Proc.
proc means data=work.bank;

run;

proc print data=bank;

run;

proc contents data=bank;

run;

proc means data=sasdata2.baseball;

run;
We will be discussing a number of SAS Procs throughout the workshop.

